import std.algorithm;
import std.array;
import std.conv;
import std.math;
import std.range;
import std.stdio;
import std.string;
import std.typecons;

void scan(T...)(ref T a) {
    string[] ss = readln.split;
    foreach (i, t; T) a[i] = ss[i].to!t;
}
T read(T)() { return readln.chomp.to!T; }
T[] reads(T)() { return readln.split.to!(T[]); }
alias readint = read!int;
alias readints = reads!int;

long modPow(long x, long k, long m) {
    if (k == 0) return 1;
    if (k % 2 == 0) return modPow(x * x % m, k / 2, m);
    return x * modPow(x, k - 1, m) % m;
}

const MOD = 10^^9 + 7;

long calc(string s) {
    long ans = 0;
    int one = 0; // 1 の個数
    for (int i = 0; i < s.length; i++) {
        if (s[i] == '1') {
            // s[i] = 0 とする
            // 範囲 [0, i) の各桁の 1 の組み合わせは (0, 1), (1, 0) の 2 通り
            // 範囲 [i + 1, $) の各桁の組み合わせは (0, 0), (0, 1), (1, 0) の 3 通り
            long a = modPow(2, one, MOD);
            long b = modPow(3, cast(int)s.length - i - 1, MOD);
            ans = (ans + (a * b)) % MOD;
            one++;
        }
    }

    // 上の for ループで i 番目を 0 にしたが、どの桁の 1 も 0 にしない場合
    ans = (ans + modPow(2, one, MOD)) % MOD;
    return ans;
}

// 桁 DP による解法
long calc2(string s) {
    void madd(T)(ref T a, in T b) { a = (a + b) % MOD; }

    int n = cast(int)s.length;
    auto dp = new long[][](100010, 2); // dp[見た桁数][L 以下か]
    dp[0][0] = 1;
    for (int i = 0; i < n; i++) {
        if (s[i] == '0') {
            // L を維持: 選択肢は (0, 0) のみ
            madd(dp[i + 1][0], dp[i][0]);
            // L 未満: 選択肢は (0, 0), (1, 0), (0, 1) の 3 通り
            madd(dp[i + 1][1], dp[i][1] * 3);
        }
        else {
            // L を維持: 選択肢は (1, 0), (0, 1) の 2 通り
            madd(dp[i + 1][0], dp[i][0] * 2);
            // i 桁目を 0 にして L 未満へ遷移
            madd(dp[i + 1][1], dp[i][0]);
            // L 未満: 選択肢は (0, 0), (1, 0), (0, 1) の 3 通り
            madd(dp[i + 1][1], dp[i][1] * 3);
        }
    }

    return (dp[n][0] + dp[n][1]) % MOD;
}

void main() {
    string s = read!string;
    writeln(calc2(s));
}
